
Affichage du nom de l’utilisateur en fond d’écran 
& Photo de profil Windows 

1 - Utilisation de l’outil 
 

KoXo Label est un outil qui permet de d’afficher le nom indiqué sur le fond d’écran en haut à gauche. Une 
photo (ou logo) peut aussi être fournie, elle est alors affichée en dessous du nom et peut également être associée 
au profil de l’utilisateur connecté. KoXo Label est en principe appelé d’un script VBS qui indique les options à utiliser. 
 

KoXo Label est appelé avec les messages (Avec par exemple le nom de l’utilisateur) suivis éventuellement 
de la taille, la couleur du texte et la couleur du fond. KoXo Label est aussi capable d’afficher directement le nom de 
session avec le nom de l’ordinateur utilisé. 
 

Utilisation :  
 

KoXoLabel "Jean DUPONT;18;Red;White" "Comptabilité;10;Green;White" ... 
 

Chaque message peut être accompagné de 3 options séparées par un point-virgule : La taille, La couleur du 
texte et la couleur du fond. 
 

On peut ensuite ajouter les options suivantes :  
 

/?   Affiche l’aide. 

/Wallpaper=NomFichier;MéthodeAffichage 

Indique le nom du fichier de papier peint et optionnellement la 

méthode à utiliser pour l’afficher : Tiled, Centered, Stretched. 

Les formats supportés sont : BMP, JPEG ou PNG. 

/IDPicture=NomFichier;TailleX;TailleY 

Indique le nom de l’image à afficher en vignette sous le texte en 

haut à droite de l’écran. Les formats supportés sont : JPEG ou PNG. 

La hauteur par défaut est de 180 pixels. 

/RigthMargin=Valeur 

Indique la valeur de la marge droite à utiliser pour l’écriture du 

texte. Par défaut la valeur est de 10 pixels. 

/TopMargin=Valeur Indique la valeur de la marge haute à utiliser pour l’écriture du 

texte. Par défaut la valeur est de 5 pixels. 

/NoShadow Indique que l’on ne souhaite pas utilise d’ombre pour le texte ou 

l’image. La couleur de l’ombre est #303030 (Un gris foncé). La largeur 

de l’ombre pour le texte est calculée et est de 1 pixel mini et 4 

pixels pour l’image. 

/DoNotUseUserTile Indique de ne pas utiliser l’image comme photo de profil Windows. 

/DoNotDisplayIDPicture 

Indique de ne pas afficher l’image en haut à droite de l’écran, elle 

reste cependant utilisée comme photo de profil Windows. 

/BackgroundColor=Valeur 

Indique une couleur de fond uni. 

/DefaultTextColor=Valeur 

Indique une couleur par défaut pour le texte. Cette valeur est ignorée 

pour un texte dont le 4ème paramètre est indiqué. 

/Font=Police Indique une police de caractères à utiliser. La valeur par défaut 

est Arial. 

/Auto Affiche le nom de session au format compatible SAM ainsi que le nom 

de l’ordinateur sur une deuxième ligne. Ce paramètre est prioritaire 

sur les messages s’il est utilisé. (S’il s’agit d’une Session RDP, 

alors le nom de la console et du serveur sont affichés). 

2 – Précisions techniques 
 

 
Ecran d’aide 

Les 24 couleurs HTML proposées :  
 

Nom de la couleur Valeur Hex 

Beige #F5F5DC 

Black #000000 

Blue #0000FF 

Brown #A52A2A 

Cyan #00FFFF 

ForestGreen #228B22 

Gold #FFD700 

Gray #808080 

Green #008000 

Lavender #E6E6FA 

Magenta #FF00FF 

Navy #000080 

Orange #FFA500 

Peach #FFDAB9 

Pink #FFC0CB 

Purple #800080 

Red #FF0000 

Salmon #FA8072 

Silver #C0C0C0 

SkyBlue #87CEEB 

Teal #008080 

Turquoise #40E0D0 

White #FFFFFF 

Yellow #FFFF00 
 

Les couleurs peuvent aussi être indiquées au format #RRVVBB (HTML, CSS, standard RGB) ou $BBVVRR 
(Windows GDI, Delphi)  

3 – Utilisation avec KoXo Administrator 

 

Avec les modèles de KoXo Administrator, les scripts de login appellent un 
script VBS nommé « Wallpaper.VBS ». 
 

Ce script récupère les informations de l’utilisateur dans Active Directory et appelle ensuite KoXo Label 32 
bits ou 64 bits avec la ligne de commande suivante : 
 

Execute GetLogonServer + "\NETLOGON\Wallpaper.vbs  

/BackGroundColor=$A56E3A" 
 

Ou, en fonctionnement multisites : 
 

Execute GetLogonServer + "\NETLOGON\Wallpaper.vbs  

/BackGroundColor=$A56E3A /Site=""%COMPANY_SHORT_NAME%""" 
 

Attention : Depuis la version 3.0 de KoXo Label, les couleurs personnalisées sont gérées légèrement différemment, 
le préfixe « # » indique une couleur de type HTML « RRVVBB » alors que le préfixe « $ » indique une couleur de 
type Windows GDI « BBVVRR ».  En cas de migration ou de mise à jour il faudra modifier la ligne du script login VBS 
concerné en remplaçant « # » en « $ » pour retrouver la même couleur (ne pas oublier de propager le script aux 
groupe secondaires). 
 

Le script « Wallpaper.VBS » et « KoXoLabel.exe » (x86 et x64) sont stockés dans le partage NETLOGON 
grâce à l’installation automatique réalisée par KoXo Administrator au démarrage :  
 

 
Ls script « Wallpaper.VBS » appelle la bonne version (x86 ou x64) de KoXo Label avec les paramètres : 
 

"Etablissement;24" "NomComplet;18" "(GroupePrimaire / GroupeSecondaire)" 

"[NomOrdinateur]" "/IdPicture=\\Serveur\NETLOGON\Pictures\UserId.JPG" 
 

Ou en multisites :  
 

"Etablissement;24" "NomComplet;18" "(GroupePrimaire / GroupeSecondaire)" 

"[NomOrdinateur]" "/IdPicture=\\Serveur\NETLOGON\NomSite\Pictures\UserId.JPG" 
 

Si la photo de l’utilisateur n’existe pas et qu’une image nommée « logo.png » ou « logo.jpg » est trouvée 
dans le partage « NETLOGON » alors c’est ce logo qui sera affiché. 
 

Les informations « Etablissement », « NomComplet » sont lues au niveau des attributs de l’utilisateur. Les 
information « GroupePrimaire » et « GroupeSecondaire » sont issues des unités d’organisation où se trouve 
l’utilisateur. 
 

Si l’utilisateur se connecte depuis un client RDP alors c’est le nom de celui-ci qui est affiché et non celui du 
serveur RDP.  

4 – Script « Wallpaper.VBS » (Version x86 et x64) 
 

'//////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 

'(C)2006-2025 KoXo Developpement 

'March 2025   

'For x86 & x64   

'Script for KoXoLabel / Multisite version   

'Displays in the top-right corner of the screen: Company, User Full Name, User Groups, Computer   

'ID Picture support (Pictures in \\MyServer\NETLOGON\Pictures or \\MyServer\NETLOGON\SITE_NAME\Pictures)   

'- If the user's photo is found, it will be used.   

'- Otherwise, if a file named "Logo.jpg" or "Logo.png" is found in the site folder, it will be used instead.   

On Error Resume Next 

Dim objNetwork, objShell, objArgs, objDomain, objDC, objUser, objRecordSet 

Dim objConnection, objCommand, strUserName, Computer, RDP, FQDN, LDAP, Company, FullName, Title 

Dim GS, GP, Picture, LogoJpg, LogoPng, KoXoLabelEXE, KoXoLabel, Param, Site 

Dim arrFQDN, GivenName, PersonalTitle 

Set objArgs = WScript.Arguments 

Param = "" 

Site = "" 

If objArgs.Count > 0 Then 

    For i = 0 To objArgs.Count - 1 

        If InStr(objArgs.Item(i), "/Site=") = 0 Then 

            Param = Param & """" & objArgs.Item(i) & """ " 

        Else 

            Site = RemoveQuotes(GetValue(objArgs.Item(i))) 

        End If 

    Next 

End If 

Set objNetwork = WScript.CreateObject("WScript.Network") 

strUserName = objNetwork.UserName 

Set objShell = WScript.CreateObject("WScript.Shell") 

Computer = objShell.ExpandEnvironmentStrings("%CLIENTNAME%") 

If Computer = "%CLIENTNAME%" Or Computer = "" Then Computer = objNetwork.ComputerName 

Set objDomain = GetObject("LDAP://rootDse") 

objDC = objDomain.Get("DefaultNamingContext") 

Set objConnection = CreateObject("ADODB.Connection") 

objConnection.Open "Provider=ADsDSOObject;" 

Set objCommand = CreateObject("ADODB.Command") 

objCommand.ActiveConnection = objConnection 

objCommand.CommandText = "<LDAP://" & objDC & ">;(&(objectCategory=User)(sAMAccountName=" & strUserName & 

"));distinguishedName;subtree" 

Set objRecordSet = objCommand.Execute 

If objRecordSet.RecordCount = 1 Then 

    objRecordSet.MoveFirst 

    FQDN = objRecordSet.Fields("distinguishedName").Value 

    LDAP = "LDAP://" & FQDN 

    Set objUser = GetObject(LDAP) 

    Company = objUser.Get("company") 

    GivenName = objUser.Get("givenName") 

    FullName = objUser.Get("sn") 

    If IsNull(GivenName) Then GivenName = "" 

    If IsNull(FullName) Then FullName = "" 

    FullName = Trim(GivenName & " " & FullName) 

    PersonalTitle = objUser.Get("personalTitle") 

    If IsNull(PersonalTitle) Or PersonalTitle = "" Then 

        Title = objUser.Get("title") 

    Else 

        Title = PersonalTitle 

    End If 

    If Not IsNull(Title) And Title <> "" Then FullName = Title & " " & FullName 

    If FullName = "" Then FullName = objUser.Get("displayName") 

    If FullName = "" Then FullName = strUserName 

    arrFQDN = Split(FQDN, ",") 

    If UBound(arrFQDN) >= 1 Then GS = GetValue(arrFQDN(1)) Else GS = "?GS?" 

    If UBound(arrFQDN) >= 2 Then GP = GetValue(arrFQDN(2)) Else GP = "?GP?" 

    If Site <> "" Then Site = Site & "\" 

    Picture = Replace(WScript.ScriptFullName, WScript.ScriptName, Site & "Pictures\" & strUserName & ".jpg") 

    LogoJpg = Replace(WScript.ScriptFullName, WScript.ScriptName, Site & "Logo.jpg") 

    LogoPng = Replace(WScript.ScriptFullName, WScript.ScriptName, Site & "Logo.png") 

    ' Vérification des fichiers d'image 

    If Not FileExists(Picture) Then 

        If FileExists(LogoJpg) Then 

            Picture = "/IDPicture=" & Chr(34) & LogoJpg & ";0;180" & Chr(34) 

        ElseIf FileExists(LogoPng) Then 

            Picture = "/IDPicture=" & Chr(34) & LogoPng & ";0;180" & Chr(34) 

        Else 

            Picture = "" 

        End If 

    Else 

        Picture = "/IDPicture=" & Chr(34) & Picture & Chr(34) 

    End If 

    Param = Param & """" & FullName & ";18"" ""(" & GP & " / " & GS & ")"" ""[" & Computer & "]""" 

    If Company <> "" Then Param = """" & Company & ";24"" " & Param 

    If OsIs64bits() Then 

        KoXoLabelEXE = "KoXoLabel64.exe" 

    Else 

        KoXoLabelEXE = "KoXoLabel32.exe" 

    End If 

    KoXoLabel = Replace(WScript.ScriptFullName, WScript.ScriptName, KoXoLabelEXE) 

    objShell.Run KoXoLabel & " " & Param & " " & Picture, 1, True 

End If 

objConnection.Close 

WScript.Quit 

'//////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 

'Get Value from ValueName=Value 

Function GetValue(Val) 

    Dim i 

    i = InStr(Val, "=") 



Utilisation de KoXo Label V3 
http://www.koxo.net 

© 2006-2025 KoXo Développement 
 

6 

    If i > 0 Then 

        GetValue = Mid(Val, i + 1) 

    Else 

        GetValue = "" 

    End If 

End Function 

'//////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 

'Remove framing quotes 

Function RemoveQuotes(Val) 

    l = Len(Val) 

    If l > 2 Then 

        If (Mid(Val, 1, 1) = Chr(34)) And (Mid(Val, l, 1) = Chr(34)) Then 

            RemoveQuotes = Mid(Val, 2, l - 2) 

        Else 

            RemoveQuotes = Val 

        End If 

    Else 

        RemoveQuotes = Val 

    End If 

End Function 

'//////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 

'Check if a file exists 

Function FileExists(FilePath) 

    Dim fso 

    Set fso = CreateObject("Scripting.FileSystemObject") 

    FileExists = fso.FileExists(FilePath) 

    Set fso = Nothing 

End Function 

'//////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 

'64 bits ? 

Function OsIs64bits() 

    Dim objWMIService, colOperatingSystems, objOperatingSystem, OSArchi 

    Set objWMIService = GetObject("winmgmts:\\.\root\cimv2") 

    Set colOperatingSystems = objWMIService.ExecQuery("Select * from Win32_OperatingSystem") 

    OSArchi = "" 

    For Each objOperatingSystem In colOperatingSystems 

        If Not IsNull(objOperatingSystem.OSArchitecture) Then OSArchi = objOperatingSystem.OSArchitecture 

    Next 

    OsIs64bits = (InStr(OSArchi, "64") > 0) 

End Function 

'//////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 

 

 
 


